

【EM アルゴリズム (スカラー) EM Algorithm-scalar-】

x を未知量, v をノイズ, y を観測値とし, Model: $y = x + v$ に従うとする.
また, 確率分布を以下のように仮定する.

未知量 x の事前確率分布

$$f(x|a) = \mathcal{N}(x|0, a^{-1}) = \sqrt{\frac{a}{2\pi}} \exp\left(-\frac{a}{2}x^2\right)$$

観測値の確率分布

$$f(y|x, b) = \mathcal{N}(y|x, b^{-1}) = \sqrt{\frac{b}{2\pi}} \exp\left(-\frac{b}{2}(y-x)^2\right)$$

事後確率分布

$$f(x|y) = \mathcal{N}(x|\bar{x}, c^{-1}) = \sqrt{\frac{c}{2\pi}} \exp\left(-\frac{c}{2}(x-\bar{x})^2\right)$$

E-Step

$$c = a + b, \quad \bar{x} = \frac{b}{a + b}y$$

M-Step

ハイパーパラメータ a, b の更新式

$$\hat{a} = \frac{1}{\bar{x}^2 + c^{-1}}, \quad \hat{b} = \frac{1}{(y - \bar{x})^2 + c^{-1}}$$




Figure 1: Scilab 実行結果

Source Code 1: Scilab

```
////////////////////////////////////  
//      EMアルゴリズム(スカラー)  
//      EM Algorithm-scalar-  
//
```

```

//                               M. Tsutsui
////////////////////////////////////

clear;

d_size=100;//データサイズ

t=linspace(0,3*%pi,d_size);

x=-sin(2*t)+1.5*cos(3*t)-2*sin(5*t);//真値

v=(4*rand(1,d_size)-3*rand(1,d_size));//ノイズ

y=x+v;//観測値

uc=10;//更新回数

a=0.01;//ハイパーパラメータ a初期値
b=0.01;//ハイパーパラメータ b初期値

for i=1:1:d_size;
    for j=1:1:uc;//更新ループ
        c=a+b;//E-step

        x_temp=b/(a+b)*y(i);//E-step

        a=1/(x_temp^2+1/c);//M-step

        b=1/((y(i)-x_temp)^2+1/c);//M-step

    end
    x_bar(i)=x_temp;//推定値格納
end

plot(x);//真値
plot(y,'k--');//観測値
plot(x_bar,'r');//推定値
legend(['True';'Observation';'Estimate']);
title('EM_Algorithm-scalar-', 'fontsize',4);

```